# Maintainer: George Rawlinson <george@archlinux.org>
# Contributor: matoro <throwaway19587@gmail.com>
# Contributor: Patrick Burroughs (Celti) <celti@celti.name>

pkgname=html-xml-utils
pkgver=8.5
pkgrel=1
pkgdesc='Utilities for manipulating HTML and XML files'
arch=('loong64' 'x86_64')
url='https://www.w3.org/Tools/HTML-XML-utils/'
license=('W3C')
depends=('curl' 'libidn2')
source=("$url/$pkgname-$pkgver.tar.gz")
sha512sums=('f1140a81e66cc2f46dd0507a8948b2c267230fa8aed3dc79e91eeb2b6bcfe7053378da9d35fd053f8a8bc25c46491388aa7fd519c7e1c7757e73a7e1b1c67d87')
b2sums=('8fa959104db2d5765ceaf49be815f16c1422a3c7b94e0e8cd91e3bd0a8b18dfc33dff5d98204d7fe5694cfb711d213a5d82f58159b6d493fc78336d8dae1824d')

build() {
  cd "$pkgname-$pkgver"
  ./configure --prefix=/usr
  make
}

check() {
  cd "$pkgname-$pkgver"
  make check
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
